﻿Type=Activity
Version=3.15
@EndOfDesignText@
#Region  Activity Attributes 
	#FullScreen: False
	#IncludeTitle: True
#End Region

Sub Process_Globals
	'These global variables will be declared once when the application starts.
	'These variables can be accessed from all modules.
	
	Dim V_XM As String
	Dim V_CH As String
	Dim V_BHID As Int
	
	Dim FBHID As Int
	Dim FBHData As String
End Sub

Sub Globals

	Dim wbvBH As WebView
	Dim HttpJobTool As HttpJob
	
End Sub

Sub Activity_Create(FirstTime As Boolean)
	Activity.LoadLayout("FormBHRYInfo")
	Activity.Title = "病人基本信息-"  & V_CH & " " &V_XM 
	
	InitLayOut
	
	'RefreshData
End Sub

Sub Activity_Resume
	If FBHID<>V_BHID Then
		RefreshData
	Else
		wbvBH.LoadHtml(FBHData)
	End If
End Sub

Sub Activity_Pause (UserClosed As Boolean)

End Sub

Sub InitLayOut

End Sub

Sub RefreshData
	Dim sURL,sSession,sText,sBHID As String
	
	sSession = PublicVar.GV_DesTool.Base64encrypt(PublicVar.GV_UserSession)
	
	sBHID = PublicVar.GV_DesTool.Base64encrypt(V_BHID)
	
	sText = "Session=" & sSession & "&BHID=" & sBHID

	sURL = PublicVar.GetServerUrl&"/getylbhinfo"
	HttpJobTool.Initialize("Job1",Me)
	HttpJobTool.PostString(sURL,sText)

	ProgressDialogShow("正在获取数据，请稍候...")	

End Sub

Sub JobDone (Job As HttpJob)
   Log("JobName = " & Job.JobName & ", Success = " & Job.Success)
   If Job.Success = True Then
      If Job.JobName = "Job1" Then
				HttpJobSuccess(Job)
      End If
   Else
		HttpJobError(Job)
   End If
   Job.Release
End Sub

Sub HttpJobSuccess(Job As HttpJob)   
	Dim sResult,sCode,sData As String
	Dim mapResult As Map
	Dim lsRecords As List
	
	ProgressDialogHide
	
	sResult = Job.GetString.Trim
	
	PublicVar.GV_JSonTool.Initialize(sResult)
	mapResult = PublicVar.GV_JSonTool.NextObject
	
	sCode = mapResult.Get("Code")
	If sCode<>"0" Then
		sResult = mapResult.Get("Message")
		ToastMessageShow("获取数据失败"&CRLF &sResult, True)	
		
		PublicVar.CheckSession(sCode)
		Return
	End If
	
	sData = mapResult.Get("Data")
	If sData ="" Then
		ToastMessageShow("获取数据失败"&CRLF &"返回的数据为空", True)	
		Return	
	
	End If
	
	sData = PublicVar.GV_DesTool.Base64decrypt(sData)
	If wbvBH.IsInitialized Then
		wbvBH.LoadHtml(sData)
		FBHData = sData
		FBHID = V_BHID
	End If

End Sub

Sub HttpJobError(Job As HttpJob)  
	Dim sMsg As String
	ProgressDialogHide
	
	Log("Error: " & Job.ErrorMessage)
	sMsg = "获取病人基本信息失败."& CRLF & Job.ErrorMessage 
  ToastMessageShow(sMsg, True)
End Sub

